@ echo off
:: 本程序用来开关无线站点
:: Author:K.T.
:: Date:2013/1/5
:: Modify:2013/12/20 [v2]
::        2015/9/8   [v3]

::====================================== 编辑下面（设定名称ssid和密码[至少8位]passwd） ===
set ssid=lbbwifi
set passwd=hanxue2015
::====================================== 编辑上面（设定名称ssid和密码[至少8位]passwd） ===


color 0b
echo.
echo           Enjoy the program: Windows无线热点配置脚本 ^^^^
echo                                                    By K.T.

echo.
echo 无线热点名称：【 %ssid% 】 
echo.
echo  [1]  Set WiFi. 设定WiFi热点的名称和使用密码。首次配置WiFi时使用该选项。
echo  [2]* Start Wlan WiFi  #回车默认该选项
echo  [3]  Stop Wlan WiFi

echo.

set /p Choice=Choose[2]：

if '%Choice%' == ''  goto START
if '%Choice%' == '2' goto START
if '%Choice%' == '3' goto STOP
if '%Choice%' == '1' goto SET
echo.
echo. Error choice.
echo.
pause


goto END

:START
netsh wlan start hostednetwork
goto SUCCESS

:STOP
netsh wlan stop hostednetwork
goto SUCCESS

:SET

echo 使用该选项，请确保为管理员权限！若否请退出并右击该文件，选择“以管理员身份运行”
echo.
echo 热点名称(ssid=):   %ssid%
echo 使用密码(passwd=): %passwd%
echo.

echo 确认上述结果？若更改请退出程序并右键点击该文件选择“编辑”。
pause

netsh wlan set hostednetwork ssid=%ssid% key=%passwd%

set /p Choice=是否已经将正使用的网络共享给了该WiFi热点？(1*:是；2:否)，如果不清楚请选否。[1] 
if '%Choice%' == ''  goto SUCCESS
if '%Choice%' == '1' goto SUCCESS 
if '%Choice%' == '2' goto DETAIL
echo.
echo. Error choice.
echo.
pause
goto END

:DETAIL
echo 请仔细看好弹出的消息。并按其说明一步一步操作。
netsh wlan start hostednetwork

msg %username%  本地WiFi站点已经设置，它的SSID是 %ssid%。
msg %username%  在本机的“网络连接”中，该WiFi站点可能显示为“本地网络*2/*3”等，它的类别是 “Microsoft 托管网络适配器”或“ Microsoft virtual WiFi miniport”。请记住这个名字，下面配置会用到。
msg %username%  下面开始4步操作，完成最后的配置。
msg %username%  (1/4) 右键点击电脑右下角的联网图标，打开网络和共享中心
msg %username%  (2/4) 在右上部找到“访问类型”为internet的网络，点击它下面的连接（以太网、本地连接、宽带连接等）[win8、win10为以太网]
msg %username%  (3/4) 在弹出的窗口中点击左下角的属性
msg %username%  (4/4) 点击“共享”选项卡，选中两个复选框，并将“家庭网络连接”选为刚配置的WiFi站点,就是刚才让你记住的那一个，点击“确定”保存配置！
msg %username%  配置完成。点击本窗口的确定后，尽情享受刚配置好的WiFi热点 %ssid% 吧^^^^

goto SUCCESS

:SUCCESS
echo 操作完成! ==========注意看上面的文字当中有没有否定的反馈！！========！！！
pause

:END